# Copyright (c) 2011 Nokia Corporation.

TARGET = qtgameenablerqmlaudiotest
TEMPLATE = app
VERSION = 2.0

CONFIG += qt warn_on

INSTALL_DIRS = qml

#DEFINES += GE_DEBUG

include(../../qtgameenabler/qtgameenableraudioplugin.pri)

SOURCES += main.cpp

OTHER_FILES += ./qml/*.qml

RESOURCES += qtgameenablerqmlaudiotest.qrc

symbian {
    TARGET = QtGameEnablerQmlAudioTest
    TARGET.UID3 = 0xee67bae1
    TARGET.EPOCSTACKSIZE = 0x14000
    TARGET.EPOCHEAPSIZE = 0x020000 0x800000

    RESOURCES += rsc/symbian.qrc

    ICON = ../../qtgameenabler/icons/qtgameenabler.svg

    contains(SYMBIAN_VERSION, Symbian3) {
        # Uses volume hack in GE audio engine to set the volume higher.
        #DEFINES += QTGAMEENABLER_USE_VOLUME_HACK

        # Define the preprocessor macro to get the version in our app.
        DEFINES += APP_VERSION=\"$$VERSION\"
    }
    else {
        message(Symbian^1)
        DEFINES += Q_OS_SYMBIAN_1
    }

    # To lock the application to portrait orientation
    LIBS += -lcone -leikcore -lavkon
}

# Unix based platforms
unix:!symbian {
    # Common
    BINDIR = /opt/usr/bin
    DATADIR = /usr/share

    target.path = $$BINDIR

    icon64.files += icons/$${TARGET}.png
    icon64.path = $$DATADIR/icons/hicolor/64x64/apps

    maemo5 {
        # Maemo 5 specific
        desktopfile.files += qtc_packaging/debian_fremantle/$${TARGET}.desktop
        desktopfile.path = $$DATADIR/applications/hildon

        RESOURCES += rsc/symbian.qrc

        INSTALLS += \
            desktopfile \
            icon64
    }
    else {
        contains(DEFINES, MEEGO_EDITION_HARMATTAN) {
            # Harmattan specific
            desktopfile.files += qtc_packaging/debian_harmattan/$${TARGET}.desktop
            desktopfile.path = $$DATADIR/applications

            DEFINES += APP_VERSION=\\\"$$VERSION\\\"

            # Classify the application as a game to support volume keys on Harmattan.
            gameclassify.files += qtc_packaging/debian_harmattan/$${TARGET}.conf
            gameclassify.path = /usr/share/policy/etc/syspart.conf.d

            RESOURCES += rsc/harmattan.qrc

            INSTALLS += \
                desktopfile \
                gameclassify \
                icon64
        }
        else {
            # Unix based desktop specific
            target.path = /usr/local/bin

            RESOURCES += rsc/symbian.qrc
        }
    }

    # Common cont.
    INSTALLS += target
}

win32 {
    RESOURCES += rsc/symbian.qrc
}


# End of file.

